

Additional Applications of SEM-based Fixed Effects Modeling with Panel Data

In this document, I expand on topics relevant to fixed effects modeling with panel data. These include modeling predictor lags, autoregression in the outcomes, autoregression in both predictors and outcomes, cross-lagged modeling, and contemporaneous reciprocal causality. I also describe how to extend SEM-based fixed effects analysis to binary, ordinal and count outcomes. Finally, I discuss multiple indicator latent variables and sensitivity tests for the biasing effects of measurement error. I assume you have read Chapter 16 and mastered the material on SEM-based fixed effects modeling within it.

Lagged Predictors in SEM-Based Fixed Effects Models

The numerical example in the main text linked the mediators MA and MB to the outcome Y vis-a-vis contemporaneous effects. Suppose I want to, in addition to the contemporaneous effects, allow for a first order lagged effect for the MB mediator. The idea is that in addition to an “immediate” MB effect on Y, there also is a delayed, independent carry over effect on Y from MB at the prior time period. Although one would not expect this to be the case for the numerical example in the main text, there are instances in some domains where lagged effects might occur. [Figure 1](#) presents an abbreviated influence diagram at a given time point that highlights the causal dynamics for just the observed variables to provide a flavor of what I am modeling. The dashed line is a short hand way of signifying that all of the exogenous variables are correlated rather than drawing the traditional pairwise curved arrows. This reduces clutter in the diagram. Note that MB has two variants, MB at time t to reflect the concurrent influence and MB at time $t-1$ to represent the lagged influence of MB on Y. Of course, the full influence diagram would include all the relevant time points and the latent α variable. [Table 1](#) presents the relevant Mplus syntax.




FIGURE 1. Model for lagged effect of MB

Table 1: Mplus Syntax for Lagged Predictor

```

1.  TITLE: Fixed effect analysis with lagged MB
2.  DATA: FILE = FEmainM.dat ;
3.  VARIABLE:
4.  NAMES ARE id za zb cov0 cov1 cov2 cov3 cov4 ma0 ma1 ma2 ma3 ma4
5.      mb0 mb1 mb2 mb3 mb4 y0 y1 y2 y3 y4 treat covmean mamean mbmean ymean ;
6.  USEVARIABLES = ma1 ma2 ma3 ma4 mb0 mb1 mb2 mb3 mb4
7.      cov1 cov2 cov3 cov4 y1 y2 y3 y4 ;
8.  ANALYSIS: ESTIMATOR=MLR ;
9.  MODEL:
10.  alpha BY y1@1 y2@1 y3@1 y4@1;
11.  y1 ON ma1 mb1 mb0 cov1 (p1 p2 p3 b1) ;
12.  y2 ON ma2 mb2 mb1 cov2 (p1 p2 p3 b1) ;
13.  y3 ON ma3 mb3 mb2 cov3 (p1 p2 p3 b1) ;
14.  y4 ON ma4 mb4 mb3 cov4 (p1 p2 p3 b1) ;
15.  y1 y2 y3 y4 (evar) ;
16.  alpha WITH ma1-ma4 mb0-mb4 cov1-cov4 ;
17.  OUTPUT: Samp StdYX Mod(All 4) Residual Tech4 ;

```

Noteworthy in this syntax is the omission of the line predicting Y at time 0 and the omission of Y0 from Lines 10 and 15. This is because we cannot include the lagged predictor for MB for Y0 as a measure of MB before time 0 does not occur in the data set. The disadvantage of using a lagged predictor is that you lose some information in the model. I also had to alter the USEVARIABLES line relative to the corresponding line in the main text to reflect only those variables in the model lines of Table 1. Some researchers might decide to not include Y1 in the model because the MB lagged variable in this case represents a baseline measure that likely has a different time lag between it and the concurrent MB as compared to the

other time points. However, as I discussed in the main text, for this particular example, the differing time lags likely are irrelevant given the substantive content of the variables. Nevertheless, the inclusion of baseline information is an issue to think about. An alternative strategy to dealing with the differential time lags is to remove the equality constraint for the path linking MB0 to Y1 or for that matter, all of the paths linking the lagged MB to Y. When I re-ran the analysis using only Y2 through Y4 and freeing up the equality constraints, the results were comparable to what I report here.

The global fit indices of the model from the syntax in [Table 1](#) all suggested satisfactory model fit. The chi square was 32.98 with $df=43$, $p < 0.87$; the RMSEA was <0.001 with a 90% confidence interval of 0.00 to 0.012; the p value for close fit was < 1.00 ; the CFI was 1.00 and the standardized RMR was 0.008. For localized fit, there were no statistically significant differences between the predicted and observed covariances on a cell-by-cell basis and there were no theoretically meaningful modification indices above 4.0.

Here are the results for the key path coefficients:

MODEL RESULTS

| | | Two-Tailed | | | |
|----|----|------------|--------|-----------|---------|
| | | Estimate | S.E. | Est./S.E. | P-Value |
| Y1 | ON | MA1 | 0.497 | 0.011 | 44.341 |
| | | MB1 | 0.490 | 0.012 | 40.563 |
| | | MB0 | -0.003 | 0.012 | -0.249 |
| | | COV1 | 0.201 | 0.012 | 16.182 |
| Y2 | ON | MA2 | 0.497 | 0.011 | 44.341 |
| | | MB2 | 0.490 | 0.012 | 40.563 |
| | | MB1 | -0.003 | 0.012 | -0.249 |
| | | COV2 | 0.201 | 0.012 | 16.182 |
| Y3 | ON | MA3 | 0.497 | 0.011 | 44.341 |
| | | MB3 | 0.490 | 0.012 | 40.563 |
| | | MB2 | -0.003 | 0.012 | -0.249 |
| | | COV3 | 0.201 | 0.012 | 16.182 |
| Y4 | ON | MA4 | 0.497 | 0.011 | 44.341 |
| | | MB4 | 0.490 | 0.012 | 40.563 |
| | | MB3 | -0.003 | 0.012 | -0.249 |
| | | COV4 | 0.201 | 0.012 | 16.182 |

Because of the equality constraints I built into the analysis, the results are identical for each time point. The results for the contemporaneous MA and MB closely follow the results in the main text where the lagged MB predictor was not included. The coefficient for the lagged MB predictor was near zero ($-.003 \pm 0.02$) and statistically non-significant (CR = -0.249, ns). In this case, the lagged predictor could reasonably be dropped from the model. Had its effect been meaningful, the model fit indices in the original analysis likely would have been amiss. The equality constraints I imposed in this analysis could be relaxed, per my discussion in the main text.

Lagged Outcomes in SEM-Based Fixed Effects Models

The numerical example in the main text did not include lagged causal effects of prior Y on current Y . The assumption was that any correlation between the Y across time were largely due to the common causes of time-invariant variables or the correlations among the predictors across time. If I believed there were meaningful within-person first order autoregressive causal effects for the Y , then I would want to add causal paths representing these effects. [Figure 2](#) presents an abbreviated influence diagram at a given time point that highlights the presumed causal dynamics for just the observed variables to provide a flavor of what I am modeling. Of course, the full influence diagram would include all the relevant time points and the latent α variable. [Table 2](#) presents the relevant Mplus syntax.




FIGURE 2. Model for first order autoregressive effect for Y

Table 2: Mplus Syntax for Autoregressive Y Effects

```

1.  TITLE: Fixed effect analysis with autoregressive Y
2.  DATA: FILE = FEmainM.dat ;
3.  VARIABLE:
4.    NAMES ARE id za zb cov0 cov1 cov2 cov3 cov4 ma0 ma1 ma2 ma3 ma4
5.      mb0 mb1 mb2 mb3 mb4 y0 y1 y2 y3 y4 treat covmean mamean mbmean ymean ;
6.  USEVARIABLES = ma1 ma2 ma3 ma4 mb1 mb2 mb3 mb4 cov1 cov2 cov3 cov4
7.      y0 y1 y2 y3 y4 ;
8.  ANALYSIS: ESTIMATOR=MLR ;
9.  MODEL:
10.    alpha BY y1@1 y2@1 y3@1 y4@1; !define latent y
11.    y1 ON ma1 mb1 y0 cov1 (p1 p2 p3 b1) ; !regress t1 y onto t1 preds
12.    y2 ON ma2 mb2 y1 cov2 (p1 p2 p3 b1) ; !regress t2 y onto t2 preds
13.    y3 ON ma3 mb3 y2 cov3 (p1 p2 p3 b1) ; !regress t3 y onto t3 preds
14.    y4 ON ma4 mb4 y3 cov4 (p1 p2 p3 b1) ; !regress t4 y onto t4 preds
15.    y1 y2 y3 y4 (evar);           !estimate disturbance variances
16.    alpha WITH ma1-ma4 mb1-mb4 cov1-cov4 y0 ; !correlate latent var with all
17.  OUTPUT: Samp StdYX Mod(All 4) Residual Tech4 ;

```

As before, I omit a line that predicts Y at time 0 and omit Y0 from Lines 10 and 15. I also altered the USEVARIABLES line relative to the corresponding line in the main text to reflect only those variables formally included in the model lines of [Table 2](#). Also, as before, some researchers might decide to not include Y1 in the model because the lagged Y in this case represents a baseline measure in the RET. Or, instead they might consider removing the equality constraint for the path linking Y0 to Y1 or, for that matter, the equality constraints for all of the Y autoregressive coefficients. It turns out that when I re-ran the analysis using only Y2 through Y4 and yet again removing the equality constraints, the results were comparable to the analysis I report here.

This model assumes that the causal flow of the predictors to Y is unidirectional, i.e., that the mediators are free of any reverse impact by Y either concurrently or at prior time points. If Y impacts the mediators in a non-trivial way, then special accommodations need to be taken into account to deal with what is known as sequential exogeneity; see Allison et al., (2017) and Leszczensky and Wolbring (2022).

The global fit indices of the model based on the syntax in [Table 2](#) all suggest satisfactory model fit. The chi square was 29.92 with df=43, $p < 0.94$; the RMSEA was <0.001 with a 90% confidence interval of 0.00 to 0.006; the p value for close fit was < 1.00 ; the CFI was 1.00 and the standardized RMR was 0.007. For localized fit, there were no statistically significant differences between the predicted and observed covariances on a cell-by-cell basis and there were no theoretically meaningful modification indices above 4.0.

Here are the results for the key path coefficients:

MODEL RESULTS

| | | Estimate | S.E. | Two-Tailed | |
|------|----|----------|-------|------------|---------|
| | | | | Est./S.E. | P-Value |
| Y1 | ON | | | | |
| MA1 | | 0.496 | 0.011 | 44.238 | 0.000 |
| MB1 | | 0.490 | 0.012 | 42.041 | 0.000 |
| Y0 | | -0.006 | 0.013 | -0.488 | 0.626 |
| COV1 | | 0.200 | 0.012 | 16.095 | 0.000 |
| Y2 | ON | | | | |
| MA2 | | 0.496 | 0.011 | 44.238 | 0.000 |
| MB2 | | 0.490 | 0.012 | 42.041 | 0.000 |
| Y1 | | -0.006 | 0.013 | -0.488 | 0.626 |
| COV2 | | 0.200 | 0.012 | 16.095 | 0.000 |
| Y3 | ON | | | | |
| MA3 | | 0.496 | 0.011 | 44.238 | 0.000 |
| MB3 | | 0.490 | 0.012 | 42.041 | 0.000 |
| Y2 | | -0.006 | 0.013 | -0.488 | 0.626 |
| COV3 | | 0.200 | 0.012 | 16.095 | 0.000 |
| Y4 | ON | | | | |
| MA4 | | 0.496 | 0.011 | 44.238 | 0.000 |
| MB4 | | 0.490 | 0.012 | 42.041 | 0.000 |
| Y3 | | -0.006 | 0.013 | -0.488 | 0.626 |
| COV4 | | 0.200 | 0.012 | 16.095 | 0.000 |

The results for the contemporaneous MA and MB closely follow the results in the main text where the first order autoregressive Y effect was not included. The autoregressive coefficient for Y was near zero ($-.006 \pm 0.03$) and statistically non-significant (CR = -0.488, ns). In this case, the lagged Y could reasonably be dropped from the model. Had its effect been meaningful but not modeled, the global fit indices in the original analysis reported in the main text likely would have been amiss.

Parenthetically, some methodologists include Y lags to deal with what is known as **exchangeability**, namely the assumption that the correlations between Y at any two time points are the same no matter the length of the time lag between them. This assumption can be (but is not always) unrealistic with the magnitude of correlations decreasing as the length of the time lag increases (per the simplex structure discussed in the main text). Including lagged Y is a strategy for dealing with this phenomena even if one does not truly believe that inertia effects exist. Rather it is a methodological strategy for modeling exchangeability violations. Of course, if you believe inertia effects exist, you will want to include the lags.

Binary and Count Outcomes

In some RETs with multiple follow-ups the outcome is binary rather than continuous. A common strategy for dealing with this is to use logistic or probit models. Consider an RET that has no baseline assessments (in many RCTs with binary outcomes, baseline measures are not collected). Individuals are randomly assigned to a treatment or control condition. Suppose there are five posttreatment assessments, including a binary outcome (e.g., the individual achieved a satisfactory level of adherence for assigned exercise protocols, 0 = no, 1 = yes), a binary mediator (MA), a continuous mediator (MB) and a binary covariate (COV). The Mplus syntax for an SEM-based fixed effects model of the $M \rightarrow Y$ links is quite similar to the case of continuous outcomes which I elaborated in the main text in Table 16.5. [Table 3](#) presents the relevant syntax for the binary outcome RET.

Table 3: Mplus Syntax for Binary Outcome

```

1.  TITLE: Fixed effect analysis with autoregressive Y
2.  DATA: FILE =binarydvM.dat;
3.  VARIABLE:
4.    NAMES = id y1-y5 ma1-ma5 cov1-cov5 mb1-mb5;
5.    USEVARIABLES ARE y1-y5 ma1-ma5 cov1-cov5 mb1-mb5;
6.    CATEGORICAL = y1-y5;
7.  ANALYSIS: ESTIMATOR=ML;
8.  MODEL:
9.    alpha BY y1-y5@1;
10.   y1-y5 PON ma1-ma5 (a);
11.   y1-y5 PON mb1-mb5 (b);
12.   y1-y5 PON cov1-cov5 (c);
13.   alpha ON ma1-ma5 mb1-mb5 cov1-cov5;
14.  OUTPUT: Samp StdYX Tech4 ;

```

I use Mplus syntax shortcuts to make programming more efficient. First, on Lines 4-6 I refer to multiple Mplus variables that have the same name but that are differentiated by successive integers at the end of the variable name. Thus y1-y5 refers to the variables y1, y2, y3, y4 and y5, which are the outcomes measured at each of the five time periods. Line 6 tells Mplus that y1-y5 should be treated as categorical outcomes. Mplus determines internally if the measures are binary. If any of the y1-y5 contain more than two values, it is treated by Mplus as ordinal rather than binary. On Line 9, I use the variable shorthand with the `BY` command to place arrows from the latent alpha to each of the outcomes. The `@1` notation applies to each referenced arrow thereby fixing each of the five paths to the value 1.0. On Line 10, I use the `PON` shortcut. The five variables to the right of `PON` are regressed onto the five predictors to the right of `PON` but using pairwise logic. Thus, line 9 actually represents five

different ON statements, $y1 \text{ ON } ma1$; $y2 \text{ ON } ma2$; $y3 \text{ ON } ma3$; $y4 \text{ ON } ma4$; and $y5 \text{ ON } ma5$; The label at the end of the line (a) is applied to each statement thereby setting an equality constraint to the five coefficients on the line. The presence of the binary outcomes means I cannot request modification indices or the RESIDUAL option on the OUTPUT line. On Line 13, with continuous outcomes I would use the WITH command, per the main text. For technical reasons I will not sidetrack on, for the case of binary outcomes I need to use ON instead of WITH to establish an association between alpha and each of the exogenous predictors.

When I execute the syntax, because of the binary outcomes I do not obtain any of the traditional fit indices. Here is the output for global model fit:

MODEL FIT INFORMATION

| | |
|---------------------------|----|
| Number of Free Parameters | 24 |
|---------------------------|----|

Loglikelihood

| | |
|----------|-----------|
| H0 Value | -3416.657 |
|----------|-----------|

Information Criteria

| | |
|--|----------|
| Akaike (AIC) | 6881.314 |
| Bayesian (BIC) | 7002.476 |
| Sample-Size Adjusted BIC ($n^* = (n + 2) / 24$) | 6926.244 |

I obtain the classic information criteria indices as well as the model log likelihood and the number of free parameters. None of these indices, alone or in combination, are adequate as indices of model fit but I make use of them shortly to provide some perspectives on fit.

Here are the results for the log odds path coefficients linking M→Y as well as the transformed coefficients in the form of odds ratios:

MODEL RESULTS

| Y1 | ON | Estimate | S.E. | Two-Tailed | |
|------|----|----------|-------|-------------|---------|
| | | | | Est. / S.E. | P-Value |
| MA1 | | 0.407 | 0.152 | 2.687 | 0.007 |
| MB1 | | -0.243 | 0.038 | -6.362 | 0.000 |
| COV1 | | 0.338 | 0.112 | 3.023 | 0.003 |

| | | Two-Tailed | | | |
|------|----|------------|-------|-----------|---------|
| | | Estimate | S.E. | Est./S.E. | P-Value |
| Y2 | ON | | | | |
| MA2 | | 0.407 | 0.152 | 2.687 | 0.007 |
| MB2 | | -0.243 | 0.038 | -6.362 | 0.000 |
| COV2 | | 0.338 | 0.112 | 3.023 | 0.003 |
| Y3 | ON | | | | |
| MA3 | | 0.407 | 0.152 | 2.687 | 0.007 |
| MB3 | | -0.243 | 0.038 | -6.362 | 0.000 |
| COV3 | | 0.338 | 0.112 | 3.023 | 0.003 |
| Y4 | ON | | | | |
| MA4 | | 0.407 | 0.152 | 2.687 | 0.007 |
| MB4 | | -0.243 | 0.038 | -6.362 | 0.000 |
| COV4 | | 0.338 | 0.112 | 3.023 | 0.003 |
| Y5 | ON | | | | |
| MA5 | | 0.407 | 0.152 | 2.687 | 0.007 |
| MB5 | | -0.243 | 0.038 | -6.362 | 0.000 |
| COV5 | | 0.338 | 0.112 | 3.023 | 0.003 |

LOGISTIC REGRESSION ODDS RATIO RESULTS FOR OBSERVED VARIABLES

| | | 95% C.I. | | | |
|------|----|----------|-------|------------|------------|
| | | Estimate | S.E. | Lower 2.5% | Upper 2.5% |
| Y1 | ON | | | | |
| MA1 | | 1.503 | 0.228 | 1.116 | 2.022 |
| MB1 | | 0.784 | 0.030 | 0.727 | 0.845 |
| COV1 | | 1.402 | 0.157 | 1.126 | 1.745 |
| Y2 | ON | | | | |
| MA2 | | 1.503 | 0.228 | 1.116 | 2.022 |
| MB2 | | 0.784 | 0.030 | 0.727 | 0.845 |
| COV2 | | 1.402 | 0.157 | 1.126 | 1.745 |
| Y3 | ON | | | | |
| MA3 | | 1.503 | 0.228 | 1.116 | 2.022 |
| MB3 | | 0.784 | 0.030 | 0.727 | 0.845 |
| COV3 | | 1.402 | 0.157 | 1.126 | 1.745 |
| Y4 | ON | | | | |
| MA4 | | 1.503 | 0.228 | 1.116 | 2.022 |
| MB4 | | 0.784 | 0.030 | 0.727 | 0.845 |
| COV4 | | 1.402 | 0.157 | 1.126 | 1.745 |

| | | Estimate | 95% C.I. | | |
|------|----|----------|----------|------------|------------|
| | | | S.E. | Lower 2.5% | Upper 2.5% |
| Y5 | ON | | | | |
| MA5 | | 1.503 | 0.228 | 1.116 | 2.022 |
| MB5 | | 0.784 | 0.030 | 0.727 | 0.845 |
| COV5 | | 1.402 | 0.157 | 1.126 | 1.745 |

Focusing on the odds ratios, the value for MA was 1.50 with a 95% confidence interval of 1.12 to 2.02. Because the confidence interval does not contain the value of 1.0, the effect is declared statistically significant ($p < 0.05$). The lower 95% margin of error is $1.116 - 1.503 = -0.39$ and the upper 95% margin of error is $2.022 - 1.503 = 0.52$. Because MA is dichotomous and dummy coded 0, 1, the odds ratio indicates that the odds of Y when MA = 1 is 1.50 times larger than the odds of Y when MA = 0.

MB is continuous and is defined by a metric such that it has a standard deviation of approximately 1. For MB the odds ratios was 0.784 with a 95% confidence interval of 0.73 to 0.85. Because the confidence interval does not contain the value of 1.0, the effect is declared statistically significant ($p < 0.05$). The lower 95% based margin of error is $0.727 - 0.784 = -0.057$ and the upper 95% based margin of error is $0.845 - 0.727 = 0.118$. Because MB is continuous, the odds ratio indicates that for every one unit that MB increases, the odds of Y are predicted to decrease by a multiplicative constant of 0.78.

I can obtain approximate indices of these effects expressed as marginal effects in the form of probabilities and risk differences by re-running the syntax in [Table 3](#) but using the modified linear probability model (MLPM; see Timoneda, 2021; Angrist & Pischke, 2008). This requires commenting out Line 6 and changing the ON to WITH in Line 13. Here are the results for MA and MB, reported only for Y1 to save space because the results for Y2 to Y5 are identical given the equality constraints:

MODEL RESULTS

| | | Estimate | Two-Tailed | | |
|------|----|----------|------------|-----------|---------|
| | | | S.E. | Est./S.E. | P-Value |
| Y1 | ON | | | | |
| MA1 | | 0.080 | 0.027 | 2.911 | 0.004 |
| MB1 | | -0.040 | 0.006 | -6.308 | 0.000 |
| COV1 | | 0.057 | 0.019 | 2.939 | 0.003 |

For MA, the probability that Y =1 when MA=1 minus when MA = 0 is approximately 0.08 ± 0.05 . For MB, for every one unit that MB increases, the probability that Y =1 decreases by -0.04 ± 0.01 . The margins of error are approximated by double the values of the standard errors.

In the main text, I evaluated if some of the equality constraints I imposed could be relaxed by means of chi square difference testing for nested models. I can use the same strategy for binary outcomes even though no chi square statistic is reported by Mplus. The strategy is to compare nested models by means of their respective log likelihoods and the number of free parameters in each model. In the above model, I found that the log-likelihood was -3416.657 with 24 free parameters. Suppose I re-run the model but relax the constraint that the paths from alpha to Y1 through Y5 all equal 1.0, i.e., that the time invariant variables impact Y1 through Y5 to an equal extent. To relax the constraint, I change Line 9 in Table 3 to read

```
alpha BY y1@1 y2-y5;
```

I need to fix at least one of the paths to 1.0 in order for the model to be identified which I did for y1. The paths for y2 through y5 all are free to vary. The loglikelihood yielded by this model -3409.192 with 28 free parameters. I provide a program on my website called *LL difference test* that performs a significance test comparing the two models based on their loglikelihoods by transforming them to chi square statistics. In the present case, the chi square difference in fit for the two models was 14.93 with $df = 4$, $p < 0.005$. The model with the relaxed constraint fit better than the model with the equality constraint. The sample size for this comparison was large ($N = 1,151$) and when I examined the difference in the magnitude of the key parameter estimates of interest, they were trivial in magnitude. Thus, although statistically significant, imposing the equality constraint did not matter much.

Parenthetically, the referenced program on my website assumes the use of ML as opposed to MLR as the estimation algorithm for the respective models. If you use MLR, then you will want to use the program called *Scaled LL difference test* to compare the models using log likelihoods. The formulae used for these programs is provided on the home page of the Mplus website.

For count outcomes that have highly skewed data, the analysis will typically use a negative binomial strategy with maximum likelihood. The Mplus programming is similar to Table 3 but instead of Line 6 reading `CATEGORICAL = y1-y5;` it will read `COUNT = y1-y5 (nb);`. The `nb` in parentheses after the listed variables tells Mplus to use a negative binomial model. On Line 13, you can use either the `ON` or `WITH` operator. I usually add an optional line after Line 13 and before Line 14 that reads `y1-y5 (d);`. This command tells Mplus to impose an equality constraint on the dispersion parameters across time. The core output follows the same format as above but with the statistics tailored to a negative binomial model per Chapter 14.

More Complex Models

Allison et al.'s (2017) approach works well for the prior model types but cases occur where more complex causal structures between mediators and outcomes need to be tackled. I have found it more productive to shift to an alternative modeling strategy within Mplus to better deal with the complexity while preserving the spirit of SEM-based fixed effects analysis Zyphur et al. (2020a, 2020b). Consider the model in [Figure 3](#) that links two mediators, MA and MB, to an outcome Y with 3 month intervals between the time periods. To avoid clutter, I omit disturbance terms for the endogenous variables as well as the time- varying covariate COV but parameters for them will be included in the Mplus syntax I describe below. There is a fixed effect factor for each mediator and covariate as well as the outcome. In prior models we only had one latent fixed effect variable for the outcome. We need to treat the variables as having separate fixed effect factors because the mediators and covariates all have autoregressive lags. They are necessary for the programming to work.




FIGURE 3. Model with complex causal dynamics

Note that the path coefficient for the first variable that the latent factor influences does not have a measured lag in the model. The path coefficient for it is estimated rather than fixed at 1.0. By doing so, the first variable loading creates an initial impact value of the fixed effect factor on the observed variable which then remains constant throughout the series. It is a way of adjusting for the missing lag in the autoregressive chains. The model portion of the model that links MB to Y across time has the form of a classic cross-lagged panel model *after adjusting for all unmeasured time-invariant influences on MB and Y*. It allows for MB to influence Y at a later point in time and for Y to influence MB. By contrast, the portion of the model that links MA to Y assumes lagged and autoregressive effects but treats the influence of the one variable on the other as unidirectional from MA to Y but not vice versa. I did not include the covariate COV in the diagram but you should assume for this example that it has the same causal structure to Y as MA does to Y. [Table 4](#) presents the relevant Mplus syntax.

Table 4: Mplus Syntax for Complex Model

```

1.  TITLE: Fixed effect analysis of a complex model
2.  DATA: FILE = feclpm.dat;
3.  VARIABLE:
4.      NAMES ARE Y1 Y2 Y3 Y4 MA1 MA2 MA3 MA4
5.      MB1 MB2 MB3 MB4 COV1 COV2 COV3 COV4 ;
6.  ANALYSIS: ESTIMATOR=MLR;
7.  MODEL:
8.      !fixed effects latent variables and their correlations
9.      y1-y4 mal-ma4; mb1-mb4 cov1-cov4;
10.     bly BY y2-y4@1 y1 ;
11.     bma BY ma2-ma4@1 ma1 ;
12.     bmb BY mb2-mb4@1 mb1 ;
13.     bcov BY cov2-cov4@1 cov1 ;
14.     bly with bma bmb bcov;
15.     bma with bmb bcov ;
16.     bmb with bcov ;
17.     !autoregressions
18.     y2-y4 PON y1-y3 (a) ;
19.     ma2-ma4 PON mal-ma3 (b) ;
20.     mb2-mb4 PON mb1-mb3 (c) ;
21.     cov2-cov4 PON cov1-cov3 (d) ;
22.     !lagged effects
23.     y2-y4 PON mal-ma3 (e) ;
24.     y2-y4 PON mb1-mb3 (f) ;
25.     y2-y4 PON cov1-cov3 (g) ;
26.     mb2-mb4 PON y1-y3 (h) ;
27.     !contemporaneous correlated disturbances
28.     y1-y4 pwith mal-ma4 (i);

```

```

29. y1-y4 pwith mb1-mb4 (j);
30. y1-y4 pwith cov1-cov4 (k);
31. ma1-ma4 pwith mb1-mb4 (l) ;
32. ma1-ma4 pwith cov1-cov4 (m);
33. cov1-cov4 pwith mb1-mb4 (n) ;
34. MODEL INDIRECT:
35. y4 IND MB1 ;
36. OUTPUT: Samp StdYX Mod(All 4) Residual Tech4 ;

```

Much of the syntax is self-explanatory. I use the `PON` syntax structure that I introduced in Table 3. Line 9 to 13 define the fixed effect latent factors. Note that I fix the paths at times 2 to 4 to 1.0 and then estimate the first path coefficient. I do not put the variable whose path is to be estimated first because Mplus by default sets the first listed variable in a `BY` statement to 1.0. Lines 14-17 define the correlations between the latent factors while Lines 17-21 define the first order autoregressive effects in the model. I introduce labels in parentheses at the end of each line but before the semi-colon. For `PON` syntax, these labels create equality constraints for all the paths implied by the line. For example, on Line 18 the paths for $y_1 \rightarrow y_2$, $y_2 \rightarrow y_3$ and $y_3 \rightarrow y_4$ all are constrained to be equal by virtue of the specification `(a)`. There is no measured lag for y_1 , so it is omitted. Of course, you can relax the equality constraints if the results yield a poor model fit or suggest in some way that the constraints need to be modified via the modification indices or nested chi square tests as illustrated in the main text.

Lines 22 to 26 define the lagged effects in the model. Lines 27 to 33 define places in the model where I believe correlated disturbances are needed. These lines use the `PWITH` command which is similar in function to the `PON` command but it specifies covariances instead of causal paths. In this case, I specify correlated disturbances between the variables measured at the same time points. This is likely necessary because there are no contemporaneous paths in the model (setting aside the factors); they all are lagged in one form or another. In theory, I might not need these parameters once I have controlled for the time-invariant variables, but I might decide to include them based on theory. Lines 34 and 35 ask for an indirect effect analysis. This is not necessary but I include to illustrate a point I want to make later.

The global model fit indices were satisfactory. The chi square was 84.968 with $df=92$, $p < 0.69$; the RMSEA was <0.001 with a 90% confidence interval of 0.00 to 0.014; the p value for close fit was < 1.00 ; the CFI was 1.00 and the standardized RMR was 0.015. For localized fit, there were no statistically significant differences between the predicted and observed covariances on a cell-by-cell basis and there were no theoretically meaningful modification indices above 4.0.

Here are the relevant path coefficients which I only report for the second time point because the values were constrained to be equal across time points. I omit results for the covariate because they are not of substantive interest. I add some annotations in red.

MODEL RESULTS

| | | Two-Tailed | | | |
|--|-----|------------|-------|-----------|---------|
| | | Estimate | S.E. | Est./S.E. | P-Value |
| Y2 ON (effects of lagged MA and MB on Y) | | | | | |
| | MA1 | 0.219 | 0.022 | 10.168 | 0.000 |
| | MB1 | 0.235 | 0.024 | 9.695 | 0.000 |
| MB2 ON (cross-lagged effect of Y on MB) | | | | | |
| | Y1 | 0.172 | 0.022 | 7.707 | 0.000 |
| Y2 ON (autoregressive effects) | | | | | |
| | Y1 | 0.466 | 0.031 | 15.048 | 0.000 |
| MA2 ON | | | | | |
| | MA1 | 0.444 | 0.046 | 9.604 | 0.000 |
| MB2 ON | | | | | |
| | MB1 | 0.541 | 0.061 | 8.879 | 0.000 |

All of the coefficients are statistically significant, $p < 0.05$.

The MODEL INDIRECT commands can be used to explore long run effects. The commands on Line 34 and 35 yield estimates of the causal coefficient linking a one unit change in MB at time 1 on Y one year later (Y4). Here is the relevant output:

TOTAL, TOTAL INDIRECT, SPECIFIC INDIRECT, AND DIRECT EFFECTS

| | | Two-Tailed | | | |
|------------------------|----------------|------------|-------|-----------|---------|
| | | Estimate | S.E. | Est./S.E. | P-Value |
| Effects from MB1 to Y4 | | | | | |
| | Total | 0.189 | 0.035 | 5.394 | 0.000 |
| | Total indirect | 0.189 | 0.035 | 5.394 | 0.000 |
| Specific indirect 1 | | | | | |
| | Y4 | | | | |
| | Y3 | | | | |
| | Y2 | | | | |
| | MB1 | 0.051 | 0.008 | 6.174 | 0.000 |

| | | | | |
|---------------------|-------|-------|-------|-------|
| Specific indirect 2 | | | | |
| Y4 | | | | |
| Y3 | | | | |
| MB2 | | | | |
| MB1 | 0.059 | 0.011 | 5.564 | 0.000 |
| Specific indirect 3 | | | | |
| Y4 | | | | |
| MB3 | | | | |
| Y2 | | | | |
| MB1 | 0.010 | 0.003 | 3.514 | 0.000 |
| Specific indirect 4 | | | | |
| Y4 | | | | |
| MB3 | | | | |
| MB2 | | | | |
| MB1 | 0.069 | 0.019 | 3.693 | 0.000 |

The effect of a one unit change in MB at time 1 is estimated to produce a 0.189 ± 0.070 increase in Y one year later (Y4), which is statistically significant (CR = 5.39, $p < 0.05$). The Specific indirect causal effects listed beneath this total effect of MB1 on Y4 identify the different mediational chains through which MB at time 1 reaches Y at time 4. They are MB1 → Y2 → Y3 → Y4, MB1 → MB2 → Y3 → Y4, MB1 → Y2 → MB3 → Y4, and MB1 → MB2 → MB3 → Y4. The separate coefficients associated with each of these mediational chains (0.051, 0.059, 0.010, 0.069) sum together to yield the total effect of 0.189. For a more detailed discussion of analyzing long term effects in SEM-based fixed effects modeling, see Shamsollahi, Zyphur and Ozkok (2022).

Contemporaneous Reciprocal Causality

I can use the above programming strategy to estimate contemporaneous reciprocal causality in a within person sense again using SEM-based fixed effects analysis. I illustrate the programming strategy here using a simple example with only two variables, MA and Y, measured at four time points per [Figure 4](#). It is straight forward to add additional mediators and covariates. For more details on the approach (but implemented using random intercept modeling, see Speyer et al, 2025 and Muthén and Asparouhov (2024). The figure omits disturbance terms to avoid clutter but they are taken into account in the Mplus syntax.

Key to model implementation is the presence of instrumental variables for the reciprocal causal relationships as discussed in the main text. In this case, the lagged MA and lagged Y serve as instruments. For a discussion of using lagged variables as instruments, see Bellemare, Masaki and Pepinsky (2017) and Wang and Bellemare (2020). Their use requires both the independence and exclusion restriction assumptions hold (see main text).




FIGURE 4. Model with contemporaneous reciprocal causality

The Mplus syntax with equality constraints appears in [Table 5](#).

Table 5: Mplus Syntax for Contemporaneous Reciprocal Causation

```

1.  TITLE: Fixed effect analysis with contemporaneous reciprocal causes ;
2.  DATA: FILE = ferecip.dat;
3.  VARIABLE:
4.    NAMES ARE Y1 Y2 Y3 Y4 MA1 MA2 MA3 MA4 ;
5.  ANALYSIS: ESTIMATOR=MLR;
6.  MODEL:
7.    !fixed effects factors
8.    bly by y2-y4@1 y1 ;
9.    bma by ma2-ma4@1 ma1 ;
10.   bly with bma ;
11.   !autoregressive Effects:
12.   y2-y4 PON y1-y3 (a) ;
13.   ma2-ma4 PON ma1-ma3 (b) ;
14.   !reciprocal effects
15.   y2-y4 PON ma2-ma4 (c) ;
  
```

```

17. ma2-ma4 PON y2-y4 (d) ;
18. !correlation with first time period
19. ma1 WITH y1;
20. !error variances
21. ma2-ma4 (e) ;
22. y2-y4 (f) ;
23. MODEL INDIRECT:
24. y2 IND MA2 ;
25. y2 IND MA2 ;
26. OUTPUT: Samp StdYX Mod(All 4) Residual Tech4 ;

```

Most of the syntax should be self-explanatory given the prior examples in this document. The model indirect commands estimate the effects of MA \rightarrow Y and Y \rightarrow MA taking into account the looping effects inherent in contemporaneous reciprocal causality (see the main text for elaboration).

When I fit the implied model to a simulated set of data, the global model fit indices were satisfactory. The chi square was 10.80 with df=22, $p < 0.98$; the RMSEA was <0.001 with a 90% confidence interval of 0.00 to <0.001 ; the p value for close fit was < 1.00 ; the CFI was 1.00 and the standardized RMR was 0.008. For localized fit, there were no statistically significant differences between the predicted and observed covariances on a cell-by-cell basis and there were no modification indices above 4.0.

Here are the core relevant path coefficients which I only report for the second time point because the values were constrained to be equal across time points:

MODEL RESULTS

| | | Two-Tailed | | | |
|-----|-----|------------|-------|-----------|---------|
| | | Estimate | S.E. | Est./S.E. | P-Value |
| Y2 | ON | | | | |
| | Y1 | 0.344 | 0.032 | 10.599 | 0.000 |
| MA2 | ON | | | | |
| | MA1 | 0.436 | 0.041 | 10.741 | 0.000 |
| Y2 | ON | | | | |
| | MA2 | 0.301 | 0.026 | 11.366 | 0.000 |
| MA2 | ON | | | | |
| | Y2 | 0.296 | 0.026 | 11.478 | 0.000 |

The first two set of results are the autoregressive coefficients and the second two set of results are for the reciprocal causal effects. Here are the results that take into account the looping effects:

TOTAL, TOTAL INDIRECT, SPECIFIC INDIRECT, AND DIRECT EFFECTS

| | Estimate | S.E. | Est./S.E. | Two-Tailed P-Value |
|------------------------|----------|-------|-----------|-----------------------|
| Effects from MA2 to Y2 | | | | |
| Total | 0.330 | 0.030 | 10.991 | 0.000 |
| Total indirect | 0.029 | 0.004 | 7.236 | 0.000 |
| Effects from Y2 to MA2 | | | | |
| Total | 0.325 | 0.029 | 11.113 | 0.000 |
| Total indirect | 0.029 | 0.004 | 7.329 | 0.000 |

Taking into account the looping effects, for every one unit that MA increases, the mean of Y is predicted to increase by 0.33 units (95% MOE = ± 0.06 , CR = 10.99, $p < 0.05$). Similarly, for every one unit that Y increases, the mean of MA is predicted to increase by 0.32 units (95% MOE = ± 0.06 , CR = 11.11, $p < 0.05$).

Latent Variable Models

It is reasonably straightforward to work with multiple indicators in SEM-based fixed effects models, but there are different ways of doing so. The approaches are discussed in Mulder Hamaker (2021). One approach defines indicator-specific intercepts that capture stable differences between persons coupled with occasion-specific factors that capture the within-person dynamics. The second approach creates a separate latent variable per occasion and then uses the latent variable as if they were “observed” variables in a single indicator modeling strategy. Figure 5 presents an example that predicts sleeping problems (S) from anxiety (A) and vice versa in a cross-lagged panel design.¹ The variables were measured at five time points with three indicators for each variable at each time point. The circles for the error and disturbance variances are omitted to avoid clutter. Of course, one would want to conduct longitudinal measurement invariance analyses for the respective measurement loadings to ensure reasonable functional equivalence in them (see the supplemental document on my website for Chapter 3, point number 6). The turquoise box contains the between-person factors. Mplus syntax follows directly from the material we have covered, so I do not present it here.

¹ This figure was adapted from Mulder and Hamaker (2021) based on Sedig (2020).




FIGURE 5. SEM model with indicators for latent variables to address measurement error

Sensitivity Tests for Measurement Error Bias

If you lack multiple indicators to adjust for measurement error, you can still perform sensitivity tests for the effects of measurement error on parameter estimates in your model by using the strategies discussed in the supplemental document on my website for the Resources tab of Chapter 3, point 5. Basically, you convert a given single indicator to a latent variable with a fixed factor loading of 1.0 and a fixed error variance value corresponding to the amount of unreliability you want to impose on the measure. The supplemental document for Chapter 3 discusses strategies you can use to choose reliability levels and illustrates Mplus syntax for implementing the approach.

Consider the program from the main text that applied the Allison et al. (2017) SEM-based fixed effects approach. I reproduce the Mplus syntax for it in Table 6. In Table 7, I provide the Mplus syntax that introduces 10% random error into the observed Y at each time point using the strategy outlined in the supplement. Note Lines 11 to 16 in Table 7 where I translate the observed Y into single indicator latent variables with fixed loadings of 1.0 and non-zero error variances. I then substitute those latent variables for the Y in the original analysis. To fully understand this syntax, you likely will need to first read the referenced supplement but after doing so, it should make sense to you.

Table 6: Original Syntax from Main Numerical Example in Main Text

```

1. TITLE: Evaluation of mediators on outcome
2. DATA: FILE = FEmainM.dat;
3. VARIABLE:
4. NAMES ARE id za zb cov0 cov1 cov2 cov3 cov4 ma0 ma1 ma2
5. ma3 ma4 mb0 mb1 mb2 mb3 mb4 y0 y1 y2 y3 y4 treat covmean
6. mamean mbmean ymean ;
7. USEVARIABLES ARE ma0 ma1 ma2 ma3 ma4 mb0 mb1 mb2 mb3 mb4
8. cov0 cov1 cov2 cov3 cov4 y0 y1 y2 y3 y4 ;
9. ANALYSIS: ESTIMATOR=MLR ;
10. MODEL:
11. alpha BY y0@1 y1@1 y2@1 y3@1 y4@1; !define latent time-invariant vars
12. y0 ON ma0 mb0 cov0 (p1 p2 b1) ; !regress t0 y onto t0 preds
13. y1 ON ma1 mb1 cov1 (p1 p2 b1) ; !regress t1 y onto t1 preds
14. y2 ON ma2 mb2 cov2 (p1 p2 b1) ; !regress t2 y onto t2 preds
15. y3 ON ma3 mb3 cov3 (p1 p2 b1) ; !regress t3 y onto t3 preds
16. y4 ON ma4 mb4 cov4 (p1 p2 b1) ; !regress t4 y onto t4 preds
17. y0 y1 y2 y3 y4 (evar); !estimate disturbance variances
18. alpha WITH ma0-ma4 mb0-mb4 cov0-cov4 ; !correlate latent var with all Xs
19. OUTPUT: Samp StdYX Mod(All 4) Residual Tech4 ;

```

Table 7: Original Syntax from Main Numerical Example in Main Text

```

1. TITLE: Evaluation of mediators on outcome
2. DATA: FILE = FEmainM.dat;
3. VARIABLE:
4. NAMES ARE id za zb cov0 cov1 cov2 cov3 cov4 ma0 ma1 ma2
5. ma3 ma4 mb0 mb1 mb2 mb3 mb4 y0 y1 y2 y3 y4 treat covmean
6. mamean mbmean ymean ;
7. USEVARIABLES ARE ma0 ma1 ma2 ma3 ma4 mb0 mb1 mb2 mb3 mb4
8. cov0 cov1 cov2 cov3 cov4 y0 y1 y2 y3 y4 ;
9. ANALYSIS: ESTIMATOR=MLR ;
10. MODEL:
11. !translate observed y to latent y with .9 reliability
12. ly0 BY y0@1.0 ; y0@.114 ;
13. ly1 BY y1@1.0 ; y1@.200 ;
14. ly2 BY y2@1.0 ; y2@.207 ;
15. ly3 BY y3@1.0 ; y3@.132 ;
16. ly4 BY y4@1.0 ; y4@0.125 ;
17. alpha BY ly0@1 ly1@1 ly2@1 ly3@1 ly4@1; !define latent fixed factors
18. ly0 ON ma0 mb0 cov0 (p1 p2 b1) ;
19. ly1 ON ma1 mb1 cov1 (p1 p2 b1) ;
20. ly2 ON ma2 mb2 cov2 (p1 p2 b1) ;
21. ly3 ON ma3 mb3 cov3 (p1 p2 b1) ;
22. ly4 ON ma4 mb4 cov4 (p1 p2 b1) ;
23. ly0 ly1 ly2 ly3 ly4 (evar) ; !estimate ly disturbance variances
24. alpha WITH ma0-ma4 mb0-mb4 cov0-cov4 ; !correlate latent var with all Xs
25. OUTPUT: Samp StdYX Mod(All 4) Residual Tech4 ;

```

The global fit of the model for Table 7 still yielded satisfactory fit. Here is the coefficient output for the first time point in the original model followed by the corresponding output for the model with measurement error (keep in mind that these results are identical at all time points because of the equality constraints):

MODEL RESULTS

| | | Two-Tailed | | | |
|------|----|------------|-------|-----------|---------|
| | | Estimate | S.E. | Est./S.E. | P-Value |
| Y0 | ON | | | | |
| MA0 | | 0.495 | 0.010 | 50.114 | 0.000 |
| MB0 | | 0.492 | 0.010 | 49.778 | 0.000 |
| COV0 | | 0.218 | 0.011 | 19.687 | 0.000 |

and for the second model:

| | | Two-Tailed | | | |
|------|----|------------|-------|-----------|---------|
| | | Estimate | S.E. | Est./S.E. | P-Value |
| LY0 | ON | | | | |
| MA0 | | 0.495 | 0.010 | 49.847 | 0.000 |
| MB0 | | 0.492 | 0.010 | 49.636 | 0.000 |
| COV0 | | 0.219 | 0.011 | 19.636 | 0.000 |

Note that the measurement error had little effect on the core model estimates.

Additional Applications

SEM-based fixed effects analysis can be easily pursued in conjunction with commonly used SEM analytics such as the analysis of clustered data, multiple group analyses, and bootstrapping to name a few. I illustrate such applications in future chapters of my book. Another advantage of using SEM to conduct fixed effects analysis for panel data is that it can address missing data using state of the art full information maximum likelihood, which is the default in Mplus. This is especially important for longitudinal modeling where missing data are common.

A Note on Random Effects Analyses

In the main text, I stated that the SEM-based fixed effects model assumes the within-person coefficients do not vary across individuals, i.e., that the reported coefficient for a time-varying predictor is the same for everyone. You can evaluate the reasonableness of this

assumption using the program on my website called *GLM panel regression*. In this program, you request a between-within model and follow the instructions to specify, say, from the RET example in the main text that you want the predictor MA coefficients to be allowed to vary across individuals (or what is often said “to allow for a random slope or random coefficient”). Here are the results I obtained:

WITHIN EFFECTS:

| | Est. | S.E. | t val. | d.f. | p |
|-----|-------|-------|--------|----------|-------|
| ma | 0.499 | 0.009 | 53.235 | 3979.870 | 0.000 |
| mb | 0.496 | 0.009 | 52.504 | 3997.028 | 0.000 |
| cov | 0.218 | 0.011 | 20.518 | 3997.018 | 0.000 |

RANDOM EFFECTS:

| Group | Parameter | Std. Dev. |
|----------|-------------|-----------|
| id | (Intercept) | 0.2981 |
| id | ma | 0.004372 |
| Residual | | 0.685 |

The reported within-person coefficient for MA is 0.499, which actually is the mean of the within person coefficients that were allowed to vary across individuals. In the table labeled RANDOM EFFECTS, the standard deviation across individuals of the within-person coefficients is reported and it is 0.004372. This value is extremely small and suggests that treating the within-person coefficients for MA as the same for everyone is not unreasonable.

Because the *panelr* program that underlies the R program on my website uses a different algorithm than Allison et al. (2017) to calculate the within-person coefficients with no random slopes, it usually is helpful to examine the results for such a model for comparative purposes. Here is what I found:

| | Est. | S.E. | t val. | d.f. | p |
|-------------|-------|-------|--------|----------|-------|
| (Intercept) | 0.454 | 0.023 | 19.352 | 999.000 | 0.000 |
| ma | 0.499 | 0.009 | 53.238 | 3997.000 | 0.000 |
| mb | 0.496 | 0.009 | 52.501 | 3997.000 | 0.000 |
| cov | 0.218 | 0.011 | 20.523 | 3997.000 | 0.000 |

The two sets of results are quite close which further reinforces my conclusion.

REFERENCES

Allison, P., Williams, R. & Moral-Benito, E. (2017). Maximum likelihood for cross-lagged panel models with fixed effects. *Socius*, 3, 1–7.

Angrist, J. & Pischke, J. (2008). *Mostly harmless econometrics: An empiricist's companion*. Princeton University Press.

Asparouhov, T. & Muthén, T. (2023) Residual structural equation models. *Structural Equation Modeling*, 30, 1-31.

Bellemare, M., Masaki, T. & Pepinsky, T. (2017). Lagged explanatory variables and the estimation of causal effect. *Journal of Politics*, 79,

Leszczensky, L. & Wolbring, T. (2022). How to deal with reverse causality using panel data? Recommendations for researchers based on a simulation study. *Sociological Methods and Research*, 51, 837-865.

Mulder, J. & Hamaker, E. (2021). Three extensions of the random intercept cross-lagged panel model, *Structural Equation Modeling*, 28, 638–648.

Muthén, B., & Asparouhov, T. (2024). Can cross-lagged panel modeling be relied on to establish cross-lagged effects? The case of contemporaneous and reciprocal effects.. *Psychological Methods*. Advance online publication.

Seddig, D. (2020). Individual attitudes toward deviant behavior and perceived attitudes of friends: Self-stereotyping and social projection in adolescence and emerging adulthood. *Journal of Youth and Adolescence*, 49, 664–677.

Shamsollahi, A., Zyphur, M. & Ozkok, O. (2022). Long-run effects in dynamic systems: New tools for cross-lagged panel models. *Organizational Research Methods*, 25, 435–458.

Speyer, L., Zhu, X., Yang, Y., Ribeaud, D. & Eisner, M. (2025). On the importance of considering concurrent effects in random-intercept cross-lagged panel modelling: Example analysis of bullying and internalising problems. *Multivariate Behavioral Research*, 60, 328–344.

Timoneda, J. (2021). Estimating group fixed effects in panel data with a binary dependent variable: How the LPM outperforms logistic regression in rare events data. *Social Science Research*, 93, 102486.

Wang, Y. & Bellemare, M. (2020). Lagged variables as instruments. Downloaded from <https://marcfcbellemare.com/wordpress/wp-content/uploads/2020/09/WangBellemareLagIVsJuly2020.pdf>

Zyphur, M. Allison, P., Tay, L., Voelkle, M., Preacher, K., Zhang, Z, ... & Diener, E. (2020a). From data to causes I: Building a general cross-lagged panel model (GCLM). *Organizational Research Methods*, 23, 651-687.

Zyphur, M., Voelkle, M., Tay, L., Allison, P., Preacher, K., Zhang, Z., Hamaker, E., Shamsollahi, A., Pierides, D., Koval , P. & Diener, E. (2020b). From data to causes II: Comparing approaches to panel data analysis. *Organizational Research Methods*, 23, 688–716.